Image Cover Sheet

CLASSIFICATION	SYSTEM NUMBER 149444
UNCLASSIFIED	

TITLE

AN FFT-BASED RADIATION BOUNDARY CONDITION FOR THE PARABOLIC EQUATION

System Number:

Patron Number:

Requester:

Notes:

DSIS Use only:

Deliver to: FF

Defence Research Establishment Pacific
Centre de recherches pour la défense pacifique

[redacted]

DEFENCE RESEARCH ESTABLISHMENT PACIFIC
Research and Development Branch
Department of National Defence

Canada



Contractors Report Series 94–147

AN FFT-BASED RADIATION BOUNDARY CONDITION FOR THE PARABOLIC EQUATION

Prepared by
M. ELIZABETH MAYFIELD
Department of Mathematics and Computer Science
Hood College, Frederick, MD
21701–8575 USA

Scientific Authority
Dr. David J. Thomson
Defence Research Establishment Pacific
FMO Victoria, British Columbia
CANADA V0S 1B0

Contract Serial No. W7708–2–1767/01–XSA
December 1994

I. Background

The parabolic equation (PE) model has been used very successfully for some time[1] to compute approximate solutions to the Helmholtz equation, for both range-independent and range-dependent sound propagation problems. Several finite-difference computer codes are currently in use which implement variants of this model, using efficient marching-type algorithms[2–5] to find numerical solutions to the PE.

Of particular interest here are the *external* boundary conditions associated with this equation – one for the ocean surface ($z = 0$), and one for the region below the physical ocean bottom ($z > z_b$). Traditionally, the surface has been treated as a pressure-release boundary, while a downgoing radiation condition is imposed on the field in the subbottom. To implement this latter condition numerically, an artificial absorbing layer is usually inserted below the physical bottom, and then a pressure-release condition is applied at the base of this layer. Papadakis[6] circumvented this approximate treatment by introducing an alternate *impedance* boundary condition which, though non-local, is exact and which satisfies the radiation boundary condition as $z \to \infty$. This boundary condition transforms the semi-infinite PE problem with the radiation condition at infinity to an equivalent PE problem in a bounded domain.

Implementation of this impedance condition for the standard PE involves evaluating an expression of the form[7]

$$\psi(t, z_b) = const \cdot \int_0^\infty \psi_z(\tau, z_b) \left(\int_{-\infty}^\infty \frac{\exp\left[is(t-\tau)\right] ds}{\gamma(s)} \right) d\tau. \qquad (1)$$

The inner integral in (1) is recognized to be the (inverse) Fourier transform of the function γ^{-1}. Physically, γ represents the "specific impedance" of the Fourier transform of the PE-field, namely $i\omega\rho f/f_z$, evaluated along the boundary $z = z_b$. Here f denotes the Fourier transform of ψ from t-space (range) to s-space (horizontal wavenumber), and is defined by

$$f(s, z) = \int_0^\infty \psi(t, z) \exp(-ist) dt. \qquad (2)$$

In the case of propagation over a uniform liquid half-space, the inverse Fourier transform in (1) can be evaluated analytically . But in other cases (e.g., elastic bottom, rough bottom, or layered bottom), the exact Fourier transform of the function γ^{-1} may not be known; it is then desirable to be able to evaluate this inner integral numerically. A promising FFT-based method for evaluating integrals of this type was first introduced by Mayfield.[8,9] Subsequently, this method was shown to produce accurate results provided the integrand was carefully pre-processed.[10]

The nature of the difficulty in evaluating the inner integral in 1 numerically is that, in most cases, the function γ is multivalued and exhibits at least one real zero (branch point) within the region of integration, so that the integrand has a branch cut singularity. Evaluating this integral carefully and accurately is often a difficult and non-trivial task.

In the present work, we report on some modifications to the above FFT-based approach that allow this technique to be incorporated into existing PE codes that are solved by finite-difference methods. In particular, we applied a procedure developed previously for handling similar integrals that occur in the wavenumber-integration code SAFARI.[11] The accuracy of this procedure was further improved by combining the FFT approximation to the integral with some higher-order interpolation formulas.[12,13]

II. Numerical implementation

In the case of a liquid half-space bottom, the integro-differential equation (1) assumes the specific form[7]

$$\psi(t,z_b) = \frac{\rho_b/\rho_w}{2\pi i}\int_0^\infty \psi_z(\tau,z_b)\left(\int_{-\infty}^{\infty}\frac{\exp\left[is(t-\tau)\right]ds}{\gamma(s)}\right)d\tau$$

$$= \frac{-\sqrt{i}\,(\rho_b/\rho_w)}{\sqrt{2\pi k_0}}\int_0^t \psi_z(\tau,z_b)\,\frac{\exp\left[\frac{1}{2}ik_0\left(n_b^2-1\right)(t-\tau)\right]}{\sqrt{t-\tau}}d\tau \qquad (3)$$

where the inner integral was evaluated with the aid of tables.[14] Equation 3 defines the PE-field ψ at arbitrary range t and (bottom) depth z_b in terms of an integral from 0 to t of its depth-derivative ψ_z and a kernel function that vanishes when $t<\tau$. Here ρ_b and ρ_w denote the densities of the bottom half-space $z>z_b$ and ocean $0<z<z_b$ respectively and γ is given by

$$\gamma(s) = k_0\sqrt{N_b^2-1-2s/k_0} \qquad (4)$$

where $N_b = n_b + i\alpha_b/k_0$, $n_b = c_0/c_b$ is the refractive index, and $k_0 = 2\pi f/c_0$ is an arbitrary reference wavenumber. The uniform half-space $z>z_b$ is further characterized by its sound speed c_b and attenuation α_b. In 3, the field ψ satisfies the standard PE of underwater acoustics,

$$2ik_0\frac{\partial\psi}{\partial r} + \rho\frac{\partial}{\partial z}(\rho^{-1}\frac{\partial\psi}{\partial z}) + k_0^2(N_b^2-1)\psi = 0, \qquad (5)$$

where ψ is related to the pressure p through

$$p(r,z) = \frac{\exp(ik_0r)\psi(r,z)}{\sqrt{k_0r}}. \qquad (6)$$

We are concerned in this report with the numerical evaluation of the specific inner integral I defined by

$$I(t-\tau) = \int_{-\infty}^{\infty}\frac{\exp\left[is(t-\tau)\right]ds}{\gamma(s)}. \qquad (7)$$

If we set $x = t-\tau$, $b = \frac{1}{2}k_0\left(N_b^2-1\right)$, and $B = \sqrt{ik_0/(2\pi)}$ then, from 3, the above integral may be written as

$$I(x) = \int_{-\infty}^{\infty}\frac{\exp(isx)ds}{\sqrt{2k_0}\sqrt{b-s}}$$

$$\equiv B\frac{\exp(ibx)}{\sqrt{x}} \qquad (8)$$

for $x>0$ and $I(x)=0$ for $x<0$. To evaluate I numerically, we first truncate the interval of integration between $s_{\min}<s<s_{\max}$. We take up the issue of choosing $s_{\min}$ and $s_{\max}$ later. Dividing the range of integration into $\Delta s = (s_{\max}-s_{\min})/M$ cells, then for $s_l = s_{\min}+l\Delta s$, $l = 0,1,...M$, I can be approximated by

$$I(x) = \int_{s_{\min}}^{s_{\max}} \frac{\exp(isx)ds}{\sqrt{2k_0}\sqrt{b-s}}$$

$$= \sum_{l=1}^{M} \int_{s_{l-1}}^{s_l} \frac{\exp(isx)ds}{\sqrt{2k_0}\sqrt{b-s}}$$

$$\simeq \frac{\Delta s}{\sqrt{2k_0}} \sum_{l=0}^{M-1} \frac{\exp(ixs_l)}{\sqrt{b-s_l}} \tag{9}$$

For certain values of x and M, this sum can be made into a discrete Fourier transform, or DFT, and evaluated by the fast Fourier transform (FFT) algorithm. In particular, we can choose M to be an integer power of 2, and define a set of special x's by

$$x_m \Delta s \equiv \frac{2\pi m}{M}, \quad m = 0, 1, \ldots \frac{M}{2} - 1. \tag{10}$$

Then, setting

$$f_l = \frac{\exp(ixs_l)}{\sqrt{b-s_l}} \tag{11}$$

we can write the sum for I in 9 as

$$I(x_m) \simeq \frac{\Delta s}{\sqrt{2k_0}} \exp(ix_m s_{\min}) \sum_{l=0}^{M-1} f_l \exp(2\pi i l m/M)$$

$$= \frac{\Delta s}{\sqrt{2k_0}} \exp(ix_m s_{\min}) \left[\mathrm{DFT}\,(f_0, \ldots, f_{M-1})\right]. \tag{12}$$

Press *et al.*[12,13] point out that this naive approach is not very accurate for evaluating Fourier integrals numerically and propose instead a higher-order procedure, based on combining the FFT with interpolation formulas. Replacing the DFT routine in (12) with their corrected routine DFTCOR leads to much more accurate approximations to the integral I than is possible by using a fast Fourier transform directly.

We observe that the integrand in (11) is singular at $s = b$ (and multi-valued). For lossless media ($\alpha_b = 0$), the singularity (and branch cut) lie along the real s-axis. To avoid this difficulty in implementing (12), we follow the procedure suggested in[11] and shift the contour of integration into the complex s-plane. That is, we make use of Cauchy's theorem to replace integration along the real line by integration along the shifted line $s - i\varepsilon$. For appropriate choices of $s_{\min}, s_{\max}$ and ε, we can neglect the contribution to the (contour) integral from the vertical line segments along $s = s_{\min}$ and $s = s_{\max}$. Along the shifted line $s - i\varepsilon$, the evaluation of I takes the form

$$I(x_m) \simeq \frac{\Delta s}{\sqrt{2k_0}} \exp[x_m(\varepsilon + is_{\min})] \sum_{l=0}^{M-1} g_l \exp(2\pi i l m/M)$$

$$= \frac{\Delta s}{\sqrt{2k_0}} \exp[x_m(\varepsilon + is_{\min})] \left[\mathrm{DFTCOR}\,(g_0, \ldots, g_{M-1})\right]. \tag{13}$$

where g_l is defined by

$$g_l = \frac{\exp(ixs_l)}{\sqrt{b + i\varepsilon - s_l}} \tag{14}$$

The offset amount ε is given by the prescription[11]

$$\varepsilon = \frac{3\,(s_{\max} - s_{\min})}{2\pi\,(M-1)\log e}. \tag{15}$$

and is designed to attenuate the value of the integrand by 60 dB at the maximum desired range $x_{\max}$.

For numerical work, it is necessary to select $s_{\min}$, $s_{\max}$, and M. The PE horizontal wavenumbers $s_{\min}$ and $s_{\max}$ are determined from the relationship between s and k, where k is the horizontal wavenumber associated with the solution p of the wave equation. This relationship is given by[7,15]

$$k = k_0\sqrt{1 + \frac{2s}{k_0}} \tag{16}$$

or, equivalently,

$$s = \frac{k^2 - k_0^2}{2k_0}. \tag{17}$$

To determine $k_{\min}$ and $k_{\max}$ for a given problem, we specify the phase speeds $c_{\min} = 2\pi f/k_{\max}$ and $c_{\max} = 2\pi f/k_{\min}$ where f is the frequency of the sound source. These speed limits are related to the spectral content of the propagating acoustic field. For long-range propagation, the important phase speeds are those associated with the normal modes. The choice of a reference sound speed c_0 then determines the value of $k_0 = \frac{2\pi f}{c_0}$. If c_0 is chosen to be the average speed of the propagating modes, for example, then it is seen from (17) that PE wavenumbers s corresponding to $k < k_0$ are negative.

To determine an appropriate value for M, we set the maximum range of propagation, $t_{\max}$, and the range step Δt, and choose M so that

$$\frac{1}{2}M\Delta t \geq t_{\max}.$$

Then M will be the smallest power of 2 so that $M \geq \frac{2t_{\max}}{\Delta t}$.

As noted above, use was made of the more accurate DFTCOR routine of Press *et al.*[12,13] to evaluate the Fourier integral I. In order to use this routine for a complex-valued function g, it was first necessary to modify the calling routine DFTINT (which uses routines DFTCOR, POLINT, FFOUR and REALFT) by separating g into real and complex parts and then appropriately re-combining the outputs in the calling routine. A listing of the revised routine, DFTINTC, and a program which uses this routine to find the transform of a complex-valued impedance function are attached. Note that g is defined as an external function; other problem parameters ($f, c_0, c_{\min}, c_{\max}$, and M) are entered in PARAMETER statements.

Example	f (Hz)	c_0 (m s^{-1})	$c_{\min}$ (m s^{-1})	$c_{\max}$ (m s^{-1})	M	$t_{\max}$ (km)	Δt (m)
NORDA3	250	1500	1450	1850	2048	10	2.5
Bucker	100	1500	1450	1850	2048	20	5.0

Table 1: Parameter sets for the two examples

III. Numerical Results

We present two typical examples for sound propagation over a uniform liquid half-space. In each case, we found the Fourier transform of the appropriate impedance function

$$g(s) = \frac{1}{\sqrt{\frac{k_0}{2}(N_b^2 - 1) + i\varepsilon - s}}$$

using the modified DFTCOR routine in (13) and then compared the results to the known analytic transform in (8).

An example which is often used as a test case for numerical solution of the PE is one proposed for the NORDA parabolic equation workshop.[6] A sound source of frequency 250 Hz is placed at a depth of 50 m in a water column with a constant sound speed of 1500 m s^{-1}; sound speed in the bottom is 1590 m s^{-1}. Density is 1.0 g cm^{-3} in the water and 1.2 g cm^{-3} in the bottom. Bottom attenuation is set at 0.5 dB λ^{-1}.

A second example is one proposed by Dr. Homer Bucker (see, for example[3]). In this case, the sound speed in the water column has a depth-dependent profile ranging from 1500 m s^{-1} to 1498 m s^{-1} and back to 1500 m s^{-1}; the bottom speed is 1505 m s^{-1}. The source, with frequency 100 Hz, is at a depth of 30 m; the bottom depth is 240 m. There is a jump in density from 1.0 g cm^{-3} in the water to 2.1 g cm^{-3} in the bottom.

Graphs of the real and imaginary parts of numerical and analytical transforms for each test case are attached. the plots were generated using the scientific graphic software package SAPLOT.[16]

IV. Conclusions

Significant improvements have been made in the numerical evaluation of the "inner integral" which appears in the formulation of Papadakis's non-local radiation boundary condition. First, the use of a SAFARI-like procedure to shift the contour of integration avoids the singularity in the integrand. And second, the use of a higher-order discrete Fourier transform routine with interpolation has provided a more accurate numerical approximation of the Fourier integral. This new method has been tested against analytic impedance results for two standard PE examples, with encouraging results. The next natural step will be to incorporate this improved FFT approach into one or more PE codes. Future areas for investigation include using this numerical procedure in problems with elastic or layered bottoms, or rough surfaces. The dramatic improvement in numerical results which we have seen in the completion of this project makes this a very promising approach for implementing the impedance boundary condition for the PE.



Fig. 1. Comparison of the real part of the numerical and analytical transforms for parameter set 1.



Fig. 2. Comparison of the imaginary part of the numerical and analytical transforms for parameter set 1.



Fig. 3. Comparison of the real part of the numerical and analytical transforms for parameter set 2.



Fig. 4. Comparison of the imaginary part of the numerical and analytical transforms for parameter set 2.

V. Appendix. Computer listing

The following programs are written in VAX FORTRAN Version 4.0. The routine DFTINTc is a modification of the analogous routine found in[13]; its implementation is discussed in Section II above. DFTtest tests the new numerical scheme for the specific impedance function $\gamma(s)$ given in (4).

```
      SUBROUTINE DFTINTc(FUNCc,indx,M,const,a,b,w,cosint,sinint)
c     Subroutine DFTINT, modified to transform complex-valued function
c            FUNCc

       INTEGER M,NDFT,MPOL
       REAL a,b,cosintr,cosinti,sinintr,sininti,w,TWOPI
       REAL cosint, sinint
       COMPLEX funcc,const
       PARAMETER (NDFT=8192,MPOL=6,TWOPI=2.*3.14159265)
       EXTERNAL  funcc
CU     USES dftcor,funcc,polint,realft
       INTEGER init,j,nn
       REAL aold,bold,c,cdftr,cdfti,cerr,corfacr,corfaci,corimr,corimi,
     *   correr,correi,delta,en,s,sdftr,sdfti,
     *   serr,cpolr(MPOL),cpoli(MPOL),datar(NDFT),datai(NDFT),
     *   endptsr(8),endptsi(8),spolr(MPOL),spoli(MPOL),xpol(MPOL)
       SAVE init,aold,bold,delta,datar,datai,endptsr,endptsi
       DATA init/0/,aold/-1.e30/,bold/-1.e30/
       if (init.ne.1.or.a.ne.aold.or.b.ne.bold) then
         init=1
         aold=a
         bold=b
         delta=(b-a)/M
         do 11 j=1,M+1
           datar(j)=REAL(funcc(a+(j-1)*delta,j,M,const))
           datai(j)=AIMAG(funcc(a+(j-1)*delta,j,M,const))
        write(13,*)a+(j-1)*delta,datar(j)
        write(14,*)a+(j-1)*delta,datai(j)
11       continue
         do 12 j=M+2,NDFT
           datar(j)=0.
           datai(j)=0.
12       continue
         do 13 j=1,4
           endptsr(j)=datar(j)
           endptsr(j+4)=datar(M-3+j)
           endptsi(j)=datai(j)
           endptsi(j+4)=datai(M-3+j)
13       continue
         call realft(datar,NDFT,1)
         call realft(datai,NDFT,1)
         datar(2)=0.
         datai(2)=0.
       endif
       en=w*delta*NDFT/TWOPI+1.
       nn=min(max(int(en-0.5*MPOL+1.),1),NDFT/2-MPOL+1)
       do 14 j=1,MPOL
         cpolr(j)=datar(2*nn-1)
         spolr(j)=datar(2*nn)
         cpoli(j)=datai(2*nn-1)
         spoli(j)=datai(2*nn)
         xpol(j)=nn
         nn=nn+1
14     continue
       call polint(xpol,cpolr,MPOL,en,cdftr,cerrr)
```

```
      call polint(xpol,spolr,MPOL,en,sdftr,serrr)
      call polint(xpol,cpoli,MPOL,en,cdfti,cerri)
      call polint(xpol,spoli,MPOL,en,sdfti,serri)
      call dftcor(w,delta,a,b,endptsr,correr,corimr,corfacr)
      call dftcor(w,delta,a,b,endptsi,correi,corimi,corfaci)
      cdftr=cdftr*corfacr+correr
      sdftr=sdftr*corfacr+corimr
      cdfti=cdfti*corfaci+correi
      sdfti=sdfti*corfaci+corimi
      c=delta*cos(w*a)
      s=delta*sin(w*a)
      cosintr=c*cdftr-s*sdftr
      sinintr=s*cdftr+c*sdftr
      cosinti=c*cdfti-s*sdfti
      sininti=s*cdfti+c*sdfti
      cosint=cosintr-sininti
      sinint=sinintr+cosinti
      return
      END
C  (C) Copr. 1986-92 Numerical Recipes Software
```

```
      Program DFTTest

c     Routine to test DFTCOR for 1/sqrt(b-s)
c     DFT modified to transform complex-valued function
c     *** Must be linked with FUNCc, DFTINTc, FOUR1, POLINT, REALFT,
c         DFTCOR

      Character*5     Curve
      Real kmin,kmax,k0,Littleb
      Complex Funcc,const,eye,BigB,Ref,Temp,nb

      Parameter       (Curve='Curve')
      Parameter  (eye = cmplx(0.,1.))
      Parameter (f=250.,c0=1500.,cb=1590.,ab=0.5)
      Parameter (cmin=1450.,cmax=1850.,M=2048)

c     The (possibly complex-valued) function FUNCc is imported ...
      External FUNCc

      pi=4.*atan(1.)

      k0 = 2.*pi*f/c0
      Attb = ab*f/(cb*20.*alog10(exp(1.)))
      nb = c0/cb+eye*attb/k0
      dt = 20.
      ds = 2.*pi/(M*dt)
      kmin = 2.*pi*f/cmax
      kmax = 2.*pi*f/cmin
      smin = (0.5/k0)*(kmin-k0)*(kmin+k0)
      smax = (0.5/k0)*(kmax-k0)*(kmax+k0)
      LittleB = (0.5*k0)*(nb-1.)*(nb+1.)
      eps = 3.*(smax-smin)/(2.*pi*(M-1)*alog10(exp(1.)))
      type *, 'eps = ',eps
      type *, 'smin = ',smin,'  smax = ',smax
      a = smin
      b = smax

      type *, 'Little b = ', LittleB
      const = LittleB + eye*eps

c     The following lines set up output files which can later be used
c     in the graphics software SAPLOT:

      Open    (Unit=11,File='Refftcor.sap',Status='New')
      Open    (Unit=12,File='Imfftcor.sap',Status='New')
      Open (Unit=13,File='Realinput.sap',Status='New')
      Open        (Unit=14,File='Imaginput.sap',Status='New')

      Delta_t=(b-a)/M
      Delta_w=(2.*pi)/(M*Delta_t)
      Type *, 'Delta_t = ', Delta_t
      Type *, 'Delta_w = ', Delta_w
      BigB = csqrt((2.*pi)/(eye*k0))
      type *, 'BigB = ',BigB
```

```
      Write(11,*) Curve
      Write(12,*) Curve
      Write(13,*) Curve
      Write(14,*) Curve

      Do i=1,4096
           W=i*2.5
           CALL DFTINTc(FUNCc,i,M,const,A,B,W,COSINT,SININT)
c     Real and imaginary parts of the transform are re-combined in
c     the appropriate order:
           Temp = (cosint + eye*sinint)*exp(w*eps)/sqrt(2.*k0)
           Write(11,*)     W, Real(Temp)
           Write(12,*)     W, Aimag(Temp)
      EndDo

c       Compare to Analytic Result:

      Write(11,*) Curve
      Write(12,*) Curve
      Do i=1,4096
           W=i*2.5
           Ref = BigB*cexp(eye*LittleB*w)/sqrt(w)
           write(11,*)W,Real(Ref)
           write(12,*)W,Aimag(Ref)
      EndDo

      Stop
      End
```

```
      Complex Function Funcc(s,m,MFFT,const)
c
      Complex const,eye
      Parameter (pi= 3.14159265)
      Parameter (eye = (0.,1.))
      funcc = 1./csqrt(const-s)
c
      Return
      End
```

VI. Research Bibliography

[1]F.D. Tappert, "The parabolic approximation method," in *Wave Propagation and Underwater Acoustics*, eds. J.B. Keller and J.S. Papadakis (Springer, New York, 1977), Chap V, pp. 224-287.

[2]F.D. Tappert and R.H. Hardin, "Computer Simulation of long-range acoustic propagation using the parabolic equation method," in *Proceedings of the Eighth International Congress on Acoustics*, (Goldcrest, London, 1974), Vol. 2, p. 452.

[3]D. Lee and S.T. McDaniel, "Ocean Acoustic Propagation by Finite Difference Methods," Comput. Math. Applic. **14**, 305-423 (1987).

[4]D.J. Thomson, "Wide-angle parabolic equation solutions to two range-dependent benchmark problems," J. Acoust. Soc. Am. **87**, 1514-1520 (1990).

[5]M.D. Collins, "Benchmark calculations for higher-order parabolic equations," J. Acoust. Soc. Am. **87**, 1535-1538 (1990).

[6]J.S. Papadakis, "Impedance formulation of the bottom boundary condition for the parabolic equation model in underwater acoustics," in *NORDA Parabolic Equation Workshop, 31 March - 3 April 1981*, Naval Ocean Research and Development Activity, NSTL Station, MS, Tech. Note 143, p. 83 (1982).

[7]D.J. Thomson and M.E. Mayfield, "An exact radiation condition for use with the *a posteriori* PE method," J. Comp. Acoust. **2**, 113-132 (1994).

[8]M.E. Mayfield, *Nonreflective Boundary Conditions for Schroedinger's Equation*, Ph.D. thesis, University of Rhode Island, 1989.

[9]J.S. Papadakis, M.I. Taroudakis and B. Mayfield, "Treatment of an elastic bottom in the parabolic approximation," J. Acoust. Soc. Am. Suppl. 1 **86**, S65 (1989).

[10]J.S. Papadakis, M.I. Taroudakis, P.J. Papadakis and B. Mayfield, "A new method for a realistic treatment of the sea bottom in the parabolic equation," *J. Acoust. Soc. Am.* **92**, 2030-2038 (1992).

[11]H. Schmidt," SAFARI: Seismo-Acoustic Fast Field Algorithm for Range-Independent Environments," SACLANT Undersea Research Centre, San Bartolomeo, Italy, Rep. SR-113, 1988.

[12]W.H. Press and S. A. Teukolsky, "Computing accurate integrals with the FFT," Comp. in Phys. **3**, 91-94 (1989).

[13]W.H. Press, S.A. Teukolsky, W.T. Vetterling, and B.P. Flannery, *Numerical Recipes in Fortran: The Art of Scientific Computing, 2nd Ed.* (Cambridge University Press, New York, 1992), pp. 577-584.

[14]I.S. Gradshteyn and I.M. Rhyzhik. *Tables of Integrals, Series, and Products*, ed. A. Jeffrey (Academic Press, New York, 1980), p. 318, entry No. 7.

[15] D.J. Thomson and D.H.Wood, "A postprocessing method for removing phase errors in the parabolic equation," J. Acoust. Soc. Am. **82**, 224-232 (1987).

[16] C.A. Young, "SAPLOT: Scientific graphic software user's manual," Defence Research Establishment Atlantic, Dartmouth, Nova Scotia, Tech. Commun. 91/307, 1991.

UNCLASSIFIED

SECURITY CLASSIFICATION OF FORM
(highest classification of Title, Abstract, Keywords)

DOCUMENT CONTROL DATA

(Security classification of title, body of abstract and indexing annotation must be entered when the overall document is classified)

1 ORIGINATOR (the name and address of the organization preparing the document. Organizations for whom the document was prepared, e.g. Establishment sponsoring a contractor's report, or tasking agency, are entered in section 8.)

Department of Mathematics and Computer Science
Hood College, Frederick MD 21701-8575

2. SECURITY CLASSIFICATION (overall security classification of the document including special warning terms if applicable)

UNCLASSIFIED

3. TITLE (the complete document title as indicated on the title page. Its classification should be indicated by the appropriate abbreviation (S,C or U) in parentheses after the title.)

An FFT-Based Radiation Boundary Condition for the Parabolic Equation (U)

4 AUTHORS (Last name, first name, middle initial)

Mayfield, M. Elizabeth

5 DATE OF PUBLICATION (month and year of publication of document)

December 1994

6a. NO. OF PAGES (total containing information. Include Annexes, Appendices, etc.)

15

6b. NO. OF REFS (total cited in document)

16

7 DESCRIPTIVE NOTES (the category of the document, e.g. technical report, technical note or memorandum. If appropriate, enter the type of report, e.g. interim, progress, summary, annual or final. Give the inclusive dates when a specific reporting period is covered.)

Contractors Report Series

8. SPONSORING ACTIVITY (the name of the department project office or laboratory sponsoring the research and development. Include the address.)

DREP

9a. PROJECT OR GRANT NO. (if appropriate, the applicable research and development project or grant number under which the document was written. Please specify whether project or grant)

DRDM 08

9b. CONTRACT NO. (if appropriate, the applicable number under which the document was written)

W7708-2-1767/01-XSA

10a. ORIGINATOR'S DOCUMENT NUMBER (the official document number by which the document is identified by the originating activity. This number must be unique to this document.)

CRS 94-147

10b. OTHER DOCUMENT NOS. (Any other numbers which may be assigned this document either by the originator or by the sponsor)

11. DOCUMENT AVAILABILITY (any limitations on further dissemination of the document, other than those imposed by security classification)

( ) Unlimited distribution
( ) Distribution limited to defence departments and defence contractors; further distribution only as approved
( ) Distribution limited to defence departments and Canadian defence contractors; further distribution only as approved
( ) Distribution limited to government departments and agencies; further distribution only as approved
( ) Distribution limited to defence departments; further distribution only as approved
(x) Other (please specify): DND Only

12. DOCUMENT ANNOUNCEMENT (any limitation to the bibliographic announcement of this document. This will normally correspond to the Document Availability (11). However, where further distribution (beyond the audience specified in 11) is possible, a wider announcement audience may be selected.)

Full Unlimited

UNCLASSIFIED

SECURITY CLASSIFICATION OF FORM

DCD03 2/06/87

UNCLASSIFIED

SECURITY CLASSIFICATION OF FORM

13. ABSTRACT (a brief and factual summary of the document. It may also appear elsewhere in the body of the document itself. It is highly desirable that the abstract of classified documents be unclassified. Each paragraph of the abstract shall begin with an indication of the security classification of the information in the paragraph (unless the document itself is unclassified) represented as (S), (C), or (U). It is not necessary to include here abstracts in both offical languages unless the text is bilingual).

A numerical method is presented for evaluating the Fourier transform of $1/\gamma$, where γ is the specific impedance of the acoustic field. Integrals of this type arise in the formulation of a non-local, radiation boundary condition for the parabolic equation (PE). Using a procedure developed for the wavenumber-integration model SAFARI, the contour of integration is first shifted into the complex plane to avoid a singularity in the integrand. Then a fast Fourier transform (FFT) is used to evaluate the resulting discrete approximation to the Fourier integral. This procedure is tested by comparing the numerical calculation to the known analytic Fourier transform of $1/\gamma$ for the special case of propagation over a liquid half-space.

14. KEYWORDS, DESCRIPTORS or IDENTIFIERS (technically meaningful terms or short phrases that characterize a document and could be helpful in cataloguing the document. They should be selected so that no security classification is required. Identifiers, such as equipment model designation, trade name, military project code name, geographic location may also be included. If possible keywords should be selected from a published thesaurus. e.g. Thesaurus of Engineering and Scientific Terms (TEST) and that thesaurus-identified. If it is not possible to select indexing terms which are Unclassified, the classification of each should be indicated as with the title.)

Parabolic Equation (PE)
Fourier Transform
Specific Impedance
Non-local Boundary Condition
Integration Contour
Singular Integrand
FFT
SAFARI
Finite-difference Algorithm
Liquid Half-space

UNCLASSIFIED

SECURITY CLASSIFICATION OF FORM

149444

NO. OF COPIES / NOMBRE DE COPIES: 1	COPY NO. / COPIE Nº: 1	INFORMATION SCIENTIST'S INITIALS / INITIALES DE L'AGENT D'INFORMATION SCIENTIFIQUE: DAG
AQUISITION ROUTE / FOURNI PAR ▶ DREP		
DATE ▶ 02 Feb 91		
DSIS ACCESSION NO. / NUMÉRO DSIS ▶		

DND 1158 (6-87)

National Defence / Défense nationale

PLEASE RETURN THIS DOCUMENT TO THE FOLLOWING ADDRESS:

DIRECTOR
SCIENTIFIC INFORMATION SERVICES
NATIONAL DEFENCE
HEADQUARTERS
OTTAWA, ONT. - CANADA K1A 0K2

PRIÈRE DE RETOURNER CE DOCUMENT À L'ADRESSE SUIVANTE:

DIRECTEUR
SERVICES D'INFORMATION SCIENTIFIQUES
QUARTIER GÉNÉRAL
DE LA DÉFENSE NATIONALE
OTTAWA, ONT. - CANADA K1A 0K2